Original_To:  BITNET%"wpcorp-l@ubvm"
Newsgroups: bit.listserv.wpcorp-l
Date:         Thu, 9 Apr 1992 17:32:00 EST
              <WPCORP-L@UBVM.BITNET>
From:         "Karl W. Lohner" <KLOHNER@DRUNIVAC.BITNET>
Subject:      RE: 5.0 macro to 5.1 query.

>From: "Ms. Lucia Asturias" <LUCIA@LIVERPOOL.AC.UK>

>One of our WP users on Campus used to be able to print duplex (on both
>sides of the page) using a HP Laserjet Series II  using WP 5.0.
>This macro was programmed by the person who trained me, but he's already
>left and I'm stuck with this as I don't seem to get anywhere....!

I took a look at your macro.  In WordPerfect 5.1, you might decide
that you don't even need it anymore.  To do duplex printing the
way your macro did it in 5.0, try this in 5.1:

1. Press Print (Shift-F7) and Multiple Pages (5)

2. Type:   e,1-20   and press (Enter), where 1-20 can be whatever range
   you want to print.  This will print the even pages in that range.
   You'll have to physically reverse the order of the pages yourself for
   the next step.  Make sure you only specify the range from lower page
   number to upper page number; e,20-1 will print even pages from page
   20 to the end of the document, not even pages from page 20 to 1 in
   reverse order.

3. After you've reversed the order of the pages, put them back in the
   paper bin and press Print (Shift-F7) and Multiple Pages (5) again.

4. Yup, you guessed it.  Type   o,1-20   and press (Enter) to print
   the odd pages in that range.

But you probably still want a 5.1 flavor of your macro.  Here it is.
Replace all underscores (_) with spaces.

--Karl Lohner.   klohner@drunivac.drew.edu

----DUPLEX.WPM---------clip and save-------clip and save-------------

{DISPLAY OFF}

{;}Conversion_from_WordPerfect_5.0_macro_4/9/92
    by_Karl_Lohner.__klohner@drunivac.drew.edu~

{PROMPT}Counting_pages,_please_wait.~
    {Home}{Home}{Down}
    {ASSIGN}LPage~{SYSTEM}Page~~
    {Goto}{Goto}
{PROMPT}{Del to EOL}~

{TEXT}1~Enter_first_page_number_(0_for_full_document):_~

{IF EXISTS}1~{ELSE}{QUIT}{END IF}    {;}Quit_if_no_answer.~

{IF}{VAR 1}=0~                       {;}Print_full_document~
    {ASSIGN}1~1~
    {ASSIGN}2~{VARIABLE}LPage~~
    {GO}PrintRange~
{END IF}

{TEXT}2~Enter_last_page_number_(0_to_end_of_document):_~
                                     {;}Print_range_of_pages~
{IF}{VAR 2}=0~                       {;}To_end_of_document~
    {ASSIGN}2~{VARIABLE}LPage~~
{END IF}

{IF}{VAR 2}<{VAR 1}~                 {;}Check_valid_range~
    {ASSIGN}3~{VAR 1}~               {;}Reverse_range.~
    {ASSIGN}1~{VAR 2}~
    {ASSIGN}2~{VAR 3}~
{END IF}

{IF}{VAR 2}>{VARIABLE}LPage~~
    {BELL}
    {TEXT}3~Too_many_pages,_print_from_page_{VAR 1}_to_
            end_of_document_(Y/N)?_~
    {CASE}{VAR 3}~Y~YesPrint~y~YesPrint~~
    {Screen}{Screen}
    {QUIT}
    {LABEL}YesPrint~
    {ASSIGN}2~{VARIABLE}LPage~~
{END IF}

{LABEL}PrintRange~                   {;}Do_the_printing~

{IF}{VAR 1}={VAR 2}~
    {GO}OnePage~
{END IF}

{IF}{VAR 2}%2~                       {;}Find_last_even_page~
    {ASSIGN}3~{VAR 2}-1~
{ELSE}
    {ASSIGN}3~{VAR 2}~               {;}Var_3_is_last_even_page~
{END IF}

{LABEL}PrintEven~                    {;}Print_the_even_pages~
    {IF}{VAR 3}<{VAR 1}~             {;}Greater_than_1st_page~
        {GO}OddPages~
    {END IF}
    {Print}5{VAR 3}{Enter}
    {PROMPT}Queuing_page_{VAR 3}...~ {;}Show_status~
    {WAIT}20~
    {ASSIGN}3~{VAR 3}-2~
{GO}PrintEven~

{LABEL}OddPages~
    {PROMPT}{Del to EOL}~
    {TEXT}x~Wait_for_even_pages_to_print,_
            replace_paper_in_drawer,_then_press_Enter...{BELL}~

{LABEL}PrintOdd~
    {Print}5o,{VAR 1}-{VAR 2}{Enter}

{LABEL}EndDuplex~
    {PROMPT}Odd_pages_queued_for_printing...{BELL}~
    {WAIT}20~
    {DISPLAY ON}
    {Screen}{Screen}
{QUIT}

{LABEL}OnePage~
    {Print}5{VAR 1}{Enter}
    {PROMPT}Page_queued_for_printing...{BELL}~
    {WAIT}20~
    {DISPLAY ON}
    {Screen}{Screen}
{QUIT}
{;}The_End.~
Comments: Gated by NETNEWS@AUVM.AMERICAN.EDU